

CONTOUR POTENTIAL MAPPING

A THIRUMALAI, S BIRLA SEKARAN and R KALIDOSS
Central Electrochemical Research Institute, Karaikudi - 623 006

ABSTRACT

A computer aided mapping for electrochemical engineering analysis has been developed using a micro-computer system. Softwares for linear interpolation, sorting, serialising and plotting have been developed and reported with flow charts, program listing and plotted results.

Key words: Micro computer system, equipotential mapping, electrochemical engineering.

INTRODUCTION

Automatic equipotential mapping is an important analysing software necessary for electrochemical engineering designs. It is widely used to plot the equipotential contours in the following electrochemical studies: (a) Surface potential measurement of a corroding structure like bridge, house construction etc (b) current distribution study in electrochemical impressed current protection system (c) current distribution study in electrolysis cells, metal finishing tanks, battery grids and metal anodes and (d) temperature distribution study in molten baths and furnaces. Potential, current or temperature is determined by some techniques at various nodal points of the system being analysed. Normally collected data are analysed by manual method or by computer [1-3, 5]. Manual methods are more laborious and less precise in output. Many big computer systems [2] have their own developed programs and it is costly to use with less manipulation. In this laboratory, we have developed softwares to be used with 8 bit micro-computer. Very interesting results as desired have been obtained. Brief details on the logic followed, flow charts and listing of the programs are presented below.

Statement of the problem

A two-dimensional area can be divided into a convenient linear way whereby we get a definite number of nodal points. Functional value at these nodal points are obtained by mathematical analysis [3, 4] or by experimental methods [1]. Desired equipotential contour has to be drawn by locating the points having the same potential values. Linear interpolation method is used to fix the point of equipotential lying in between the nodal points. These points have to be connected sequentially without cutting the other contour lines. Desired number of equipotential lines has to be drawn to estimate the gradient and other parameters.

Method of approach

For our analysis, we have taken the computed results of reference [3]. Battery grid structure potentials are analysed and potential distributions are plotted in figures 1 and 2. Using mathematical equations [4], the potential values are generated and equipotential lines are plotted in figures 3 and 4. Basically the involved steps are :

- Recording of potentials are nodal points either by experimental or by mathematical analysis
- Determine the maximum and minimum extreme potential values and divide the interval into the required number.
- Determine the points having the same required potential values using interpolation and store them in a file




Fig. 1: Potential distribution in a battery grid structure without a slot cut

- Sort out the data in the file so that the points can be sequentially connected without intercepting other contour lines. Store them in the plot file.
- Plot the data taking the selected file one by one.

Program listing 1 briefs the mathematical computation of potential between two equally charged lines [4]. The computed values are stored in the file PH11.DAT. While in analysis on battery grid current distribution, the numerical values at nodal points are determined by a method reported earlier [3]. Program listing 2 does the job of interpolation along rowwise, columnwise and diagonalwise at the appropriate nodal points. The estimated values are stores in the file B13T. It is sorted and loaded into HK.DAT file. Program listing 3 serialises the points to be connected sequentially.




Fig. 2 : Potential distribution in a battery grid structure with a slot cut




Fig. 4 : Potential distribution across two spherical electrodes kept at positive potentials




Fig. 3 : Potential distribution across two spherical electrodes kept at positive and negative potentials

DESCRIPTION

Program listing - 1

This is a simple routine to demonstrate the data generation. The formula used is

$$V(I,J) = (-2e \log(r_1) - 2e \log(r_2) + 20$$

where r_1 and r_2 are the distances from the charge centres to the point (I,J) . $V(I,J)$ values are stored in PH11.DAT file.

Program listing - 2

In fig. 5, the flow chart of program listing 2 is presented. Opening the above data files, extreme limits of $V(I,J)$ are determined as MAX and MIN.




Fig. 5 : Flow chart for program listing 2

Counter ND is used to count the number of points having same potential and initially it is set to zero. Required equipotential value is selected as H. The interval between consecutive nodal points are subdivided into 10 or any desired precision. The value of H is searched at all the nodal points. If it is found, they are recorded in B13T.DAT file and the counter ND is correspondingly incremented. If it lies within the nodal points, linear interpolation is done in rowwise, columnwise and diagonalwise and the counter is incremented. The located data points for a particular H value is loaded in B13T.DAT file and the number ND having H value is printed out for reference. If another H value is required similar procedure is followed and the final data is stored in another data file. They are sorted in descending order using SORT routine and sorted in HK.DAT file.

Table I: Program Listing 1

```

10 REM PROGRAM FOR POTENTIAL BETWEEN TWO EQUALLY CHARGED LINES
20 DIM POT(50,50)
30 K=0!
40 OPEN "O",#1,"PHI1.DAT"
50 E=1!
60 FOR I=0 TO 10
70 FOR J=0 TO 10
80 IF I=0 AND J=0 GOTO 90 ELSE 110
90 POT(I,J)=20!
100 GOTO 180
110 IF I=10 AND J=0 GOTO 120 ELSE 140
120 POT(I,J)=20!
130 GOTO 180
140 R1=SQR(I^2+J^2)
150 K=K+1
160 R2=SQR((10-I)^2+J^2)
170 POT(I,J)=(-2*E*LOG(R1)-2*E*LOG(R2))+20!
180 PRINT POT(I,J),J,I
190 PRINT #1,POT(I,J)
200 NEXT J
210 NEXT I
220 CLOSE #1
230 STOP
240 END

```

Table II: Program Listing 2

```

$10 REM PROGRAM FOR INTERPOLATION TO COLLECT EQUIPOTENTIAL POINTS
20 OPTION BASE 0
30 DIM X(210,15),W$(100),YP(200),XP(200)
40 OPEN "I",#1,"PHI1.DAT"
50 FOR I=1 TO 10
60 FOR J=1 TO 10
70 INPUT #1, X(I,J)
80 PRINT X(I,J);I;J
90 NEXT J
100 PRINT
110 NEXT I
120 PRINT
130 CLOSE #1
140 GOTO 1830
150 INPUT H
160 LET ND=0!
170 OPEN "O",#1,"B13T.DAT"
180 PRINT "H= "H
190 GOSUB 530
200 GOSUB 720
210 GOSUB 1110
220 CLOSE #1
230 RESET

```

```

240 PRINT "ND= "ND
250 END
260 IF (I=10 AND K=10) AND (X(I,J)=H AND X(I,K)=H) GOTO 430
270 IF X(I,J)=H GOTO 460
280 IF X(I,K)=H GOTO 430
290 P=(X(I,J)-X(I,K))/10
300 IF X(I,J)<X(I,K) GOTO 370
310 S=X(I,J)-P
320 YP(ND)=J+.1
330 XP(ND)=I
340 S=S-P
350 YP(ND)=YP(ND)+.1
360 IF S<=H GOTO 490 ELSE 340
370 S=X(I,K)+P
380 YP(ND)=K-.1
390 XP(ND)=I
400 S=S+P
410 YP(ND)=YP(ND)-.1
420 IF S<=H GOTO 490 ELSE 400
430 XP(ND)=I
440 YP(ND)=K
450 GOTO 490
460 XP(ND)=I
470 YP(ND)=J
480 GOTO 490
490 XP(ND)=XP(ND)*10
500 YP(ND)=YP(ND)*10
510 PRINT #1,USING "###.#";YP(ND);XP(ND)
520 RETURN
530 FOR I=1 TO 10
540 FOR J=1 TO 9
550 K=J+1
560 IF X(I,J)=H XOR X(I,K)=H GOTO 660
570 IF X(I,J)=H AND X(I,K)=H GOTO 600
580 IF X(I,J)<H XOR X(I,K)<H THEN 660
590 GOTO 680
600 IF J=9 AND X(I,K)=H GOTO 620
610 GOTO 660
620 YP(ND)=K*10
630 XP(ND)=I*10
640 PRINT #1,USING "###.#";YP(ND);XP(ND)
650 GOTO 680
660 ND=ND+1
670 GOSUB 260
680 NEXT J
690 PRINT
700 NEXT I
710 RETURN
720 FOR I=1 TO 9
730 FOR J=1 TO 10
740 L=I+1
750 IF X(I,J)=H XOR X(L,J)=H GOTO 790

```

```

760 IF X(I,J)=H AND X(L,J)=H GOTO 790
770 IF X(I,J)<H XOR X(L,J)<H THEN 790
780 GOTO 820
790 ND=ND+1
800 GOSUB 860
810 PRINT #1,USING "###.#";YP(ND);XP(ND)
820 NEXT J
830 PRINT
840 NEXT I
850 RETURN
860 IF (I=9 AND J=10) AND (X(I,J)=H AND X(L,J)=H) GOTO 1060
870 IF X(I,J)=H GOTO 1030
880 IF X(L,J)=H GOTO 1060
890 P=(X(I,J)-X(L,J))/10
900 IF X(I,J)<X(L,J) GOTO 970
910 S=X(I,J)-P
920 XP(ND)=I+.1
930 YP(ND)=J
940 S=S-P
950 XP(ND)=XP(ND)+.1
960 IF S<=H GOTO 1080 ELSE 940
970 S=X(L,J)+P
980 XP(ND)=L-.1
990 YP(ND)-J
1000 S=S+P
1010 XP(ND)=XP(ND)-.1
1020 IF S<=H GOTO 1080 ELSE 1000
1030 XP(ND)=I
1040 YP(ND)=J
1050 GOTO 1080
1060 XP(ND)=L
1070 YP(ND)=J
1080 YP(ND)=YP(ND)*10
1090 XP(ND)=XP(ND)*10
1100 RETURN
1110 FOR I=1 TO 9
1120 FOR J=1 TO 9
1130 K=J+1
1140 L=I+1
1150 IF X(I,J)=H XOR X(L,K)=H GOTO 1250
1160 IF X(I,J)=H AND X(L,K)=H GOTO 1250
1170 IF X(I,J)<H XOR X(L,K)<H GOTO 1250
1180 IF X(L,J)=H XOR X(I,K)=H GOTO 1540
1190 IF X(L,J)<H XOR X(I,K)<H GOTO 1540
1200 IF X(L,J)=H AND X(I,K)=H GOTO 1540
1210 NEXT J
1220 PRINT
1230 NEXT I
1240 RETURN
1250 ND=ND+1
1260 IF (K=10 AND L=10) AND (X(I,J)=H AND X(L,K)=H) GOTO 1480
1270 IF X(I,J)=H GOTO 1450

```

```

1280 IF X(L,K)=H GOTO 1480
1290 P=(X(I,J)-X(L,K))/10
1300 IF X(I,J)<X(L,K) GOTO 1380
1310 S=X(I,J)-P
1320 YP(ND)=J+.1
1330 XP(ND)=I+.1
1340 S=S-P
1350 YP(ND)=YP(ND)+.1
1360 XP(ND)=XP(ND)+.1
1370 IF S<=H GOTO 1500 ELSE 1340
1380 S=X(L,K)+P
1390 YP(ND)=K-.1
1400 XP(ND)=L-.1
1410 S=S+P
1420 YP(ND)=YP(ND)-.1
1430 XP(ND)=XP(ND)-.1
1440 IF S<=H GOTO 1500 ELSE 1410
1450 XP(ND)=I
1460 YP(ND)=J
1470 GOTO 1500
1480 XP(ND)=L
1490 YP(ND)=K
1500 XP(ND)=XP(ND)*10
1510 YP(ND)=YP(ND)*10
1520 PRINT #1,USING "### ##";YP(ND);XP(ND)
1530 GOTO 1180
1540 ND=ND+1
1550 IF (K=10 AND L=10) AND (X(I,K)=H AND X(L,J)=H) GOTO 1770
1560 IF X(I,K)=H GOTO 1770
1570 IF X(L,J)=H GOTO 1740
1580 P=(X(I,K)-X(L,J))/10
1590 IF X(I,K)<X(L,J) GOTO 1670
1600 S=X(I,K)-P
1610 YP(ND)=K-.1
1620 XP(ND)=I+.1
1630 S=S-P
1640 YP(ND)=YP(ND)-.1
1650 XP(ND)=XP(ND)+.1
1660 IF S<=H GOTO 1790 ELSE 1630
1670 S=X(L,J)+P
1680 YP(ND)=J+.1
1690 XP(ND)=L-.1
1700 S=S+P
1710 YP(ND)=YP(ND)+.1
1720 XP(ND)=XP(ND)-.1
1730 IF S<=H GOTO 1790 ELSE 1700
1740 XP(ND)=L
1750 YP(ND)=J
1760 GOTO 1790
1770 KP(ND)=I
1780 YP(ND)=K
1790 YP(ND)=YP(ND)*10

```

```

1800 XP(ND)=XP(ND)*10
1810 PRINT #1, USING "###.##"; YP(ND), XF(ND)
1820 GOTO 1210
1830 LET MAX=X(0,0)
1840 MIN=X(0,0)
1850 FOR I=0 TO 10
1860 FOR J=0 TO 10
1870 IF MAX<X(I,J) THEN MAX=X(I,J)
1880 IF MIN>X(I,J) THEN MIN=X(I,J)
1890 NEXT J
1900 NEXT I
1910 PRINT MAX,MIN
1920 GOTO 150

```

Table III: Program Listing 3

```

10 REM PROGRAM FOR SORTING AND SERIALISING
20 DIM A(200),B(200),AY(100)
30 DIM X(200),C(100),Y(200),A1(50),A2(50);YP(100),XP(100)
40 DIM RX(100),RY(100),XX(100),YY(100),AX(100)
50 LET L=0!
60 OPEN "I",#1,"HK.DAT"
70 IF EOF(1) GOTO 110
80 L=L+1
90 INPUT #1,YP(L),XP(L)
100 GOTO 70
110 PRINT "L= " L
120 FOR I=1 TO L
130 XX(I)=YP(I)
140 YY(I)=XP(I)
150 PRINT XX(I);YY(I)
160 NEXT I
170 PRINT
180 CLOSE #1
190 J=0!
200 FOR I=1 TO L
210 IF XX(I)=XX(I+1) AND YY(I)=YY(I+1) GOTO 250
220 J=J+1
230 X(J)=XX(I)
240 Y(J)=YY(I)
250 NEXT I
260 PRINT "J= " J
270 LET JK=J
280 K=0!
290 FOR I=1 TO JK
300 IF XX(I)<>XX(I+1) GOTO 330
310 K=K+1
320 GOTO 370
330 IF K=0 GOTO 370
340 K=K+1
350 C(I)=K
360 K=0!

```

```

370 NEXT I
380 MAX=C(1)
390 FOR I=1 TO JK
400 IF MAX<C(I) THEN MAX=C(I)
410 NEXT I
420 PRINT "SS= "MAX
430 SS=MAX+1
440 FOR I=1 TO JK
450 PRINT X(I),Y(I)
460 NEXT I
470 X1=0!
480 D=SQR(10^2+10^2)
490 FOR I=1 TO JK
500 A(I)=0
510 B(I)=0
520 NEXT I
530 X=X(1)
540 Y=Y(1)
550 I=1!
560 N=0!
570 P=0!
580 M=0!
590 K=0!
600 LET I=I+1
610 IF X=X(I) AND Y=Y(I) GOTO 600
620 D1=SQR((X-X(I))^2+(Y-Y(I))^2)
630 D2=SQR((X-X(I+1))^2+(Y-Y(I+1))^2)
640 PRINT D1;X;Y;X(I);Y(I),D2;X(I+1);Y(I+1)
650 IF D1<=D AND D2<=D GOTO 700
660 IF D1<=D GOTO 890
670 IF D2<=D GOTO 950
680 IF (X(I)=0 AND Y(I)=0) AND I>M GOTO 1210
690 GOTO 1070
700 IF D1<D2 GOTO 820
710 N=N+1
720 A(N)=X
730 B(N)=Y
740 N=N+1
750 A(N)=X(I+1)
760 B(N)=Y(I+1)
770 X=X(I)
780 Y=Y(I)
790 I=I+1
800 PRINT A(N),B(N)
810 GOTO 600
820 N=N+1
830 A(N)=X
840 B(N)=Y
850 X=X(I)
860 Y=Y(I)
870 PRINT A(N),B(N)
880 GOTO 600

```

```

880 N=N+1
900 A(N)=X
910 B(N)=Y
920 X=X(I)
930 Y=Y(I)
940 GOTO 1060
950 N=N+1
960 A(N)=X
970 B(N)=Y
980 X=X(I+1)
990 Y=Y(I+1)
1000 K=K+1
1010 IF K=1 GOTO 1020 ELSE 1040
1020 P=(JK-I)
1030 M=(JK+P)
1040 X(M-K)=X(I)
1050 Y(M-K)=Y(I)
1060 GOTO 600
1070 K=K+1
1080 IF K=1 GOTO 1090 ELSE 1110
1090 P=(JK-I)
1100 M=(JK+P)
1110 X(M-K)=X(I)
1120 Y(M-K)=Y(I)
1130 K=K+1
1140 IF K=1 GOTO 1150 ELSE 1170
1150 P=(JK-I)
1160 M=(JK+P)
1170 I=I+1
1180 X(M-K)=X(I)
1190 Y(M-K)=Y(I)
1200 GOTO 600
1210 FOR I=1 TO (JK+K)
1220 PRINT X(I),Y(I),M,P
1230 NEXT I
1240 N=N+1
1250 A(N)=X
1260 B(N)=Y
1270 K=0!
1280 FOR I=1 TO N
1290 IF A(I)=A(I+1) AND B(I)=B(I+1) GOTO 1330
1300 K=K+1
1310 RX(K)=A(I)
1320 RY(K)=B(I)
1330 NEXT I
1340 FOR I=1 TO K
1350 PRINT RX(I);RY(I)
1360 NEXT I
1370 PRINT
1380 X1=X1+1
1390 PRINT "No OF LOOPS= "X1
1400 IF X1=55 GOTO 1460

```

```

1410 FOR I=1 TO K
1420 X(I)=RX(I)
1430 Y(I)=RY(I)
1440 NEXT I
1450 GOTO 480
1460 OPEN "O", #1, "H1K.DAT"
1470 FOR I=1 TO K
1480 PRINT #1, USING "###.##"; RX(I); RY(I)
1490 NEXT I
1500 CLOSE #1
1510 PRINT "N=N, "K= "K
1520 STOP
1530 END

```

Program listing - 3

In fig. 6, the flow-chart of program listing 3 is shown.




Fig. 6 : Flow chart for program listing 3

Conditioning of the HK.DAT file is done by removing repetitively occurring points. Taking X-axis as reference, the number of Y ordinates associated with each X coordinate is determined. This data has been estimated to determine the Y ordinate for a particular X coordinate to be connected sequentially in case of multi Y ordinates for a X coordinate. A search window technique has been developed. A square window of diagonal length-D is located at one extreme end of grid [say $X = x(1)$ and $Y = y(1)$]. The distance D1 and D2 of two consecutive points from (X, Y) in the array are determined. If D1 and D2 lie within the window, D1 and D2 are compared and the closest points are stored in A(N) and B(N) and the reference points X and Y are approximately shifted. Similar logical steps are followed if D1 and D2 alone lie within D. If both points lie outside the window, these points are moved to stack points outside this array of points sequentially. By these methods, the equipotential points are arranged in RX(N) and RY(N)

sequentially and are stored in H1K.DAT. Using semigraph routines available in the microcomputer unit, these files are called and plotted one by one.

The software listing of the above programs are shown in Tables I to III.

RESULTS

Figures 1 to 4 display the plots of analysed results. Figure 1 shows the equipotential contours drawn on a battery grid structure with current feeding point at the right hand corner. Concentration of gradient could be visualised near the feeding end.

Figure 2 shows the equipotential contours drawn on a battery grid structure with a slot-cut and with current feeding point at the right hand corner. Improved distribution of gradient on the right hand side of the grid and increased voltage drop on the electrode could be seen.

Figure 3 shows the potential distribution across two line electrodes kept at positive and negative potentials. Steep gradient could be seen near the electrode ends.

Figure 4 shows the potential distribution between two line electrodes kept at positive potentials. The caving-in of the potential lines could be seen near the electrodes.

Manually plotted results of the same could be seen in references [3] and [4] for comparison.

CONCLUSION

From the results, it could be seen that such softwares developed in microcomputer system could be used more effectively for contour potential mapping. More resolution could be obtained by having more memory space and applying various types of interpolation techniques. The developed program is very flexible for use.

REFERENCES

1. Steel in concrete – *Electrochemistry & Corrosion Newsletter*, No. 7, January AB1 - No : ISSN 0106 2239, Danish Corrosion Centre, Denmark
2. E A Erdelyi, E F Fuchs and D H Binkley, *IEEE Trans on Power Apparatus and Systems*, 89 - 7 (1970) 1565
3. L E Vaaler, E W Broaman and H A Cuggiti – *J. Appl Electrochem* 12 (1982) 721
4. W R Smyth – *Static and dynamic electricity*, McGraw Hill Book Co., Inc., New York, 1950
5. D E Hall, E J Taylor and D J Kerstanski, *Plating & Surf Finish* 72 - 3 (1985) 60